
1 Standard Disclaimer  

The intent of this DSIC is to provide confidence that the data distributed by the NIDDK repository is a 
true copy of the study data. Our intent is not to assess the integrity of the statistical analyses reported 
by study investigators. As with all statistical analyses of complex datasets, complete replication of a set 
of statistical results should not be expected in secondary analysis. This occurs for a number of reasons 
including differences in the handling of missing data, restrictions on cases included in samples for a 
particular analysis, software coding used to define complex variables, etc. Experience suggests that most 
discrepancies can ordinarily be resolved by consultation with the study data coordinating center (DCC), 
however this process is labor-intensive for both DCC and Repository staff. It is thus not our policy to 
resolve every discrepancy that is observed in an integrity check. Specifically, we do not attempt to 
resolve minor or inconsequential discrepancies with published results or discrepancies that involve 
complex analyses, unless NIDDK Repository staff suspect that the observed discrepancy suggests that 
the dataset may have been corrupted in storage, transmission, or processing by repository staff. We do, 
however, document in footnotes to the integrity check those instances in which our secondary analyses 
produced results that were not fully consistent with those reported in the target publication.  

2 Study Background 

The RISE Pediatric Medication Study was a two-arm, multi-center clinical trial.  Study participants were 
enrolled at four recruitment centers and randomized into two treatment groups.  The first group 
received metformin alone for 12 months.  The second group received glargine, a man-made insulin, for 
three months followed by metformin for the remaining nine months of the treatment period.  Beta-cell 
function and insulin sensitivity were assessed at baseline, the end of the active treatment period, and 
three months after treatment was withdrawn.   

3 Archived Datasets  

All data files, as provided by the Data Coordinating Center (DCC), are located in the RISE Pediatric 
Medication study data package. For this replication, variables were taken from the derived datasets: 
risepeds_basedata.sas7bdat, risepeds_otherlabdata.sas7bdat, risepeds_baseline.sas7bdat, 
risepeds_clamp.sas7bdat, risepeds_clampdata.sas7bdat, risepeds_screen.sas7bdat, 
risepeds_ogttdata.sas7bdat, and risepeds_screendata.sas7bdat. 

4 Statistical Methods  

Analyses were performed to duplicate results for the data published by Sharon L. Edelstein [1] in 
Diabetes Care in 2018. To verify the integrity of the datasets, descriptive statistics were computed. 

5 Results  

For Table 1 in the publication [1], Table 1 – Baseline demographic and physical characteristics, insulin 
sensitivity, and b-cell responses from the hyperglycemic clamp and OGTT by treatment group, Table A 
lists the variables that were used in the replication and Tables B-1 to B-4 compare the results calculated 
from the archived data file to the results published in Table 1.  

6 Conclusions  

The results of the replication are within expected variation to the published results given the following 
restrictions. One subject that was included in the publication was not included in the data due to 
consent restrictions. For female subjects, race and ethnicity were coded as non-Hispanic White and all 
other race/ethnicity groups were combined. Due to a lower count of male subjects, race and ethnicity 
were not included in the data.   

7 References  

[1] Sharon L. Edelstein. “Impact of Insulin and Metformin Versus Metformin Alone on b-Cell Function in 
Youth With Impaired Glucose Tolerance or Recently Diagnosed Type 2 Diabetes” Diabetes Care 2018 
Aug; 41: 1717-1725. https://doi.org/10.2337/dc18-0787. 

Table A: Variables used to replicate Table 1 – Baseline physical and demographic characteristics by 
treatment group  

Characteristic dataset.variable 
Weight (kg) risepeds_clamp.weight                       

Age (years) risepeds_basedata_niddk_only.agerand 

BMI (kg/m2)                risepeds_baseline.weight1, risepeds_baseline.height1                      

Waist circumference (cm) risepeds_baseline.waist1 

HbA1c  risepeds_otherlabdata.hba1c                 

Hyperglycemic clamp, Fasting glucose (mmol/L)   risepeds_clampdata.gluc_05                   

Hyperglycemic clamp, Fasting C-peptide (nmol/L) risepeds_clampdata.cpep_05                   

Hyperglycemic clamp, Glucose infusion rate (M) 
(mmol/kg/min) 

risepeds_clampdata.gir                

OGTT, Fasting glucose (mmol/L) risepeds_ogttdata.gluc_05                

OGTT, 2-h glucose (mmol/L)       risepeds_ogttdata.gluc120                   

Systolic BP (mmHg)         risepeds_baseline.sbp                       

Diastolic BP (mmHg)        risepeds_baseline.dbp                       

LDL cholesterol (mmol/L)   risepeds_otherlabdata.ldl                   

HDL cholesterol (mmol/L)   risepeds_otherlabdata.hdl                   

Sex risepeds_basedata.sex                

Tanner stage V risepeds_screen.scbrst, risepeds_screen.sctestvol 

Race risepeds_basedata.race               

Glycemic Group risepeds_screendata.diabetes 

Metformin Use risepeds_screen.scmet, risepeds_runstart.rsmet 

Hypertensive risepeds_baseline.sbp, risepeds_baseline.dbp, 
risepeds_basedata_niddk_only.agerand, 
risepeds_baseline.height1, risepeds_baseline. baantihyp 

Table B-1: Comparison of values computed in integrity check to reference article Table 1 values. 
Glargine followed by metformin Cohort (N, %) 

 
    Glargine followed by metformin 
    N     %     
    Manuscript DSIC Diff Manuscript DSIC Diff 

Sex Women 27 27 0 61.4 61.4 0 
  Tanner stage V 32 32 0 72.7 72.7 0 

Race/ethnicity White 13 7 6 29.5 15.9 13.6 
  Black 14 N/A N/A 31.8 N/A N/A 
  Hispanic (any) 14 N/A N/A 31.8 N/A N/A 
  All other 3 N/A N/A 6.8 N/A N/A 

Glycemic group IGT 26 26 0 59 59.1 -0.1 
  Diabetes 18 18 0 41 40.9 0.1 

Metformin use at 
baseline Yes 10 10 0 22.7 22.7 0 

Hypertensive Yes 9 9 0 20.5 20.5 0 

Table B-2: Comparison of values computed in integrity check to reference article Table 1 values. 
Metformin Alone Cohort (N, %) 

    metformin alone 

    N     %     

    Manuscript DSIC Diff Manuscript DSIC Diff 

Sex Women 38 37 1 80.9 80.4 0.5 

  Tanner stage V 25 24 1 53.2 52.2 1 

Race/ethnicity White 12 10 2 25.5 21.7 3.8 

  Black 9 N/A N/A 19.1 N/A N/A 

  Hispanic (any) 20 N/A N/A 42.6 N/A N/A 

  All other 6 N/A N/A 12.8 N/A N/A 

Glycemic group IGT 28 27 1 60 58.7 1.3 

  Diabetes 19 19 0 40 41.3 -1.3 
Metformin use at 

baseline Yes 11 11 0 23.4 23.9 -0.5 

Hypertensive Yes 15 15 0 31.9 32.6 0.7 

 



8 

 

Table B-3: Comparison of values computed in integrity check to reference article Table 1 values. Glargine followed by metformin Cohort (Mean, 
SD) 

    Glargine followed by metformin 
    N Mean     SD     

    Manuscript DSIC Diff Manuscript DSIC Diff 
  Age (years) 44 14.9 14.86 0.04 2 2 0 
  Weight (kg) 44 102 102.06 -0.06 25.7 25.61 0.09 
  BMI (kg/m2) 44 36.5 36.53 -0.03 6.4 6.37 0.03 
  BMI percentile 44 98.4 98.35 0.05 2.5 2.5 0 
  BMI z score 44 2.3 2.33 -0.03 0.4 0.42 -0.02 
  Waist circumference (cm) 43 109.8 109.71 0.09 15.8 15.85 -0.05 
  HbA1c (%) 44 5.7 5.73 -0.03 0.6 0.6 0 
  HbA1c (mol/mmol) 44 39.2 39.17 0.03 6.6 6.55 0.05 
  Systolic BP (mmHg) 44 120.7 120.73 -0.03 7.8 7.76 0.04 
  Diastolic BP (mmHg) 44 67.6 67.64 -0.04 7.7 7.7 0 
  HDL cholesterol (mmol/L) 44 1 1.02 -0.02 0.3 0.25 0.05 
  LDL cholesterol (mmol/L) 44 2.3 2.26 0.04 0.8 0.8 0 

Hyperglycemic clamp 
variables Fasting glucose (mmol/L) 44 6.04 5.96 0.08 0.85 0.74 0.11 

  Fasting C-peptide (nmol/L) 43 1.63 1.59 0.04 0.55 0.65 -0.1 

  
Glucose infusion rate (M) 
(mmol/kg/min) 44 0.025 0.025 0 0.013 0.013 0 

OGTT variables Fasting glucose (mmol/L) 43 6 5.94 0.06 0.8 0.77 0.03 
  2-h glucose (mmol/L) 42 10.2 10.18 0.02 2.5 2.47 0.03 

Table B-4: Comparison of values computed in integrity check to reference article Table 1 values. Metformin Alone Cohort (Mean, SD) 

 
    Metformin alone 

    N Mean     SD     
    Manuscript DSIC Diff Manuscript DSIC Diff 
  Age (years) 46 13.9 13.93 -0.03 2.1 2.1 0 
  Weight (kg) 46 97.7 98.23 -0.53 23.3 23.09 0.21 
  BMI (kg/m2) 46 36.9 37.1 -0.2 6.4 6.39 0.01 
  BMI percentile 46 98.8 98.8 0 1.3 1.34 -0.04 
  BMI z score 46 2.4 2.4 0 0.3 0.34 -0.04 
  Waist circumference (cm) 46 109.6 109.85 -0.25 12.4 12.41 -0.01 
  HbA1c (%) 46 5.7 5.67 0.03 0.6 0.58 0.02 
  HbA1c (mol/mmol) 46 38.6 38.49 0.11 6.3 6.29 0.01 
  Systolic BP (mmHg) 46 119.5 119.46 0.04 8.7 8.75 -0.05 
  Diastolic BP (mmHg) 46 70.1 70.26 -0.16 7.9 7.94 -0.04 
  HDL cholesterol (mmol/L) 46 1 1.05 -0.05 0.2 0.21 -0.01 
  LDL cholesterol (mmol/L) 46 2.1 2.09 0.01 0.6 0.63 -0.03 

Hyperglycemic clamp 
variables Fasting glucose (mmol/L) 46 6.11 6.07 0.04 1.1 1.1 0 

  Fasting C-peptide (nmol/L) 45 1.82 1.8 0.02 0.58 0.54 0.04 

  
Glucose infusion rate (M) 
(mmol/kg/min) 46 0.023 0.024 -0.001 0.01 0.01 0 

OGTT variables Fasting glucose (mmol/L) 45 6.1 6.06 0.04 1.1 1.1 0 
  2-h glucose (mmol/L) 44 10.2 10.24 -0.04 2.8 2.81 -0.01 

Attachment A: SAS Code 
options nocenter validvarname=upcase fmtsearch=(formats) nofmterr; 
 
title '/prj/niddk/ims_analysis/RISE/prog_initial_analysis/RISE_Pediatric_Medication/RISE.Pediatric.dsic.sas'; 
run; 
 
* Peds primary outcome.pdf ; 
 
**********; 
* INPUT   ; 
**********; 
libname orig   '/prj/niddk/ims_analysis/RISE/private_orig_data/RISE Pediatric Medication/'; 
libname orig2  '/prj/niddk/ims_analysis/RISE/private_orig_data/RISE combined NIDDK only/';    
libname screen '/prj/niddk/ims_analysis/RISE/private_orig_data/RISE_10_30_2020/';  
libname refdir '/prj/niddk/ims_analysis/RISE/private_created_data/';  
 
************; 
* MACROS    ; 
************; 
%macro readin(lib, ds); 
  data &ds; 
    set &lib..&ds; 
  run; 
 
  proc contents data=&ds; 
  title3 "&ds"; 
  run; 
%mend; 
 
 
* produce n and %; 
%macro npercent(rownum, var, varf, subset, subsetname); 
  proc freq data=analy noprint; 
    where &subset = 1; 
    tables &var/list missing out=tbl1&subsetname; 
  run; 
 
  data tbl1&subsetname; 
    length covar covarf $100; 
    set tbl1&subsetname; 
    covar = "&var"; 
    covarf = put(&var,&varf..);   
    rownum = &rownum;  
  run; 
 
  data prnt&subsetname; 
    set prnt&subsetname tbl1&subsetname; 

  run; 
 
%mend; 
 
 
%macro univ(rownum, var, subset, subsetname); 
 
  proc univariate data=analy outtable= univ&subsetname noprint; 
    where &subset=1;  
    var &var  
       ; 
  run; 
   
  data univ&subsetname; 
    length covarf $100 _var_ $25;     
    set univ&subsetname; 
        covarf = "&subset";              
        rownum = &rownum;            
  run; 
   
  data prntuniv&subsetname; 
    set prntuniv&subsetname univ&subsetname; 
  run; 
   
%mend; 
 
 
 
*************; 
* FORMATS    ; 
*************; 
proc format; 
  value novalue 
      . = "No Value" 
  other = "   Value" 
  ; 
 
  value racegf 
  1='White'                         
  2='Other' 
  ; 
  
  value sexf 
  1='Male' 
  2='Female' 
  ; 
 
  value ynf 
    1 = 'yes' 

  .,2 = 'no' 
  ; 
 
run; 
 
 
%readin(orig, risepeds_baseline); 
%readin(orig, risepeds_basedata); 
%readin(orig, risepeds_clampdata); 
%readin(orig, risepeds_clamp); 
%readin(orig, risepeds_exitform); 
%readin(orig, risepeds_history); 
%readin(orig, risepeds_loclab); 
%readin(orig, risepeds_ogttdata); 
%readin(orig, risepeds_ogtt); 
%readin(orig, risepeds_otherlabdata); 
%readin(orig, risepeds_runend); 
%readin(orig, risepeds_runstart); 
%readin(orig, risepeds_screen); 
%readin(orig, risepeds_slpq_ped); 
%readin(orig, risepeds_smbg); 
%readin(orig, risepeds_status); 
%readin(orig, risepeds_unsched); 
%readin(orig, risepeds_visit); 
%readin(screen, risepeds_screendata);  
%readin(orig2, risepeds_basedata_niddk_only);  
 
 
proc freq data=risepeds_BASEDATA; 
  tables treatment/missing; 
  title3 "prelim freqs"; 
  run; 
 
proc freq data=risepeds_BASEDATA; 
  where treatment="GLARGINE + METFORMIN"; 
  tables AGEGROUP     RACE               SEX     /missing;    
run; 
 
proc means data=risepeds_OTHERLABDATA; 
  where visit ="BAS"; 
  var HBA1C; 
run; 
 
proc freq data=risepeds_baseline; 
  tables BAINSUL/missing; 
  tables waist1 *waist2 *waist3 /list missing; 
  format waist1  waist2  waist3 novalue.; 
run; 

proc freq data=risepeds_screen; 
  tables SCBRST/missing; 
run; 
 
 
* combine; 
proc sort data=risepeds_BASEDATA; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_OTHERLABDATA; 
  where visit="BAS"; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_baseline; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_clamp; 
  where visit="BAS"; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_clampdata; 
  where visit="BAS"; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_SCREEN; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_ogttdata; 
  where visit="BAS"; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_screendata; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_basedata_niddk_only; 
  by RISE_REPOSITORY_ID; 
run; 
 
data analy; 
  merge risepeds_basedata (in=in1 keep=rise_repository_id treatment agegroup race sex) 
        risepeds_otherlabdata (in=in2 keep=rise_repository_id hba1c  chol  hdl  ldl  trig vldl) 

        risepeds_baseline (in=in3 keep=rise_repository_id waist1 waist2 waist3 weight1 weight2 weight3  height1 height2 height3 
dbp sbp baantihyp  baliplow BAANTIHYP BAINSUL BAMETDISP) 
 risepeds_clamp          (in=in4 keep=rise_repository_id weight )         
 risepeds_clampdata      (in=in5 keep=rise_repository_id acprmax airg airmax GIR GLUC_05   
                                 rename=(GLUC_05=clamp_GLUC_05)) 
 risepeds_SCREEN         (in=in6 keep=rise_repository_id scbmi scbrst SCMET sctestvol) 
 risepeds_ogttdata (in=in7 keep=rise_repository_id GLUC_05 CPEP_05 gluc120)  
 risepeds_screendata (in=in8 keep=rise_repository_id diabetes gluc000s gluc120s) 
 risepeds_runstart       (in=in9 keep=rise_repository_id rsmet) 
 risepeds_basedata_niddk_only (in=in10 keep=rise_repository_id agerand)  
 ; 
  by rise_repository_id; 
 
  if treatment = 'GLARGINE + METFORMIN'    then treat_grp1 = 1; 
  if treatment = 'LIRAGLUTIDE + METFORMIN' then treat_grp2 = 1; 
  if treatment = 'METFORMIN'               then treat_grp3 = 1; 
  if treatment = 'PLACEBO'                 then treat_grp4 = 1; 
 
  * Conversions provided by Sharon in email 10/6/20:                 ; 
 
  *  Glucose: multiply by 0.0555    ; 
  GLUC_05mod = GLUC_05*0.0555; 
  gluc120mod = gluc120*0.0555;  
 
  clamp_GLUC_05mod = clamp_GLUC_05*0.0555;  
 
  *  Insulin: multiply by 7.462     ; 
  BAINSUL_pmol_l = BAINSUL*7.462; 
 
  *  C-peptide: divide by 3.003003    ; 
  CPEP_05_nmol_l = CPEP_05/3.003003    ;                     
 
  *  Triglycerides divide by 88.57    ; 
  trig_mmol_L  = trig/88.57; 
 
  *  HbA1c: Multiply by 10.93 and then subtract 23.5  ; 
  HbA1c_mol_mmol = HbA1c*10.93-23.5; 
 
  *  SI units for lipids is to divide by 38.67 (or multiply by 0.0259).; 
  chol_mmol_L = chol * 0.0259;   
  ldl_mmol_L = ldl   * 0.0259; 
  hdl_mmol_L = hdl   * 0.0259;  
  non_hdl_mmol_L = chol_mmol_L - hdl_mmol_L;   * Non-HDL cholesterol (mmol/L) ; 
 
  * ratio; 
  trig_hdl = trig_mmol_L/hdl_mmol_L ; 
 
  * Hypertensive     Adults were defined as hypertensive if sbp>=130 or dbp>=85 or antihyp=1 from BASELINE form; 
  if sbp>=130 or dbp>=85 or BAANTIHYP=1  then hypertensive = 1; 

  else hypertensive = 2; 
 
  *skip pattern - missing and NO grouped together;          
  if BAANTIHYP in(.,2) then BAANTIHYPmod = 2; 
  else BAANTIHYPmod = BAANTIHYP; 
  if baliplow  in(.,2) then baliplowmod  = 2; 
  else baliplowmod  = baliplow; 
 
  * Tanner stage V; 
  if (.<sctestvol<4) then male_tanner=1; 
  else if (4<=sctestvol<8) then male_tanner=2; 
  else if (8<=sctestvol<12) then male_tanner=3; 
  else if (12<=sctestvol<15) then male_tanner=4; 
  else if (sctestvol>=15) then male_tanner=5; 
   
  if sex = 2 then tanner_s5 = SCBRST; 
  else if sex = 1 then tanner_s5 = male_tanner; 
 
  if tanner_s5 = 5 then scbrst_5 = 1;        
  else if tanner_s5 ne . then scbrst_5 = 2;   
 
  *BMI; 
  bmi1 = weight1/((height1/100)**2); 
 
* calculate Glucose infusion rate (M) (mmol/kg/min) ;  
  gir_m = gir/weight1/180; 
 
run; 
 
proc freq data=analy; 
  tables treat_grp1*treat_grp2*treat_grp3*treat_grp4*treatment/list missing; 
  tables hypertensive*sbp*dbp*BAANTIHYP/list missing; 
  tables BAANTIHYPmod*BAANTIHYP/list missing; 
  tables baliplowmod*baliplow/list missing; 
  tables diabetes/missing; 
  tables scbrst_5*scbrst/list missing; 
  tables bametdisp/missing; 
  tables male_tanner*sctestvol/list missing; 
  tables tanner_s5*male_tanner*SCBRST/list missing; 
  tables scmet*rsmet/list missing; 
run; 
 
 
** create BMI z score and percentile using CDC site:  https://www.cdc.gov/nccdphp/dnpao/growthcharts/resources/sas.htm; 
data mydata (keep=RISE_REPOSITORY_ID height weight agerand agemos sex treatment bmi); 
  set analy (keep=RISE_REPOSITORY_ID agerand sex treatment bmi1 height1 weight1  
             rename=(bmi1=bmi height1=height weight1=weight)); 
 
   agemos = agerand*12; 



16 

 

 
   %include '/prj/niddk/ims_analysis/RISE/private_created_data/cdc-source-code.sas'; 
 
run; 
 
proc freq data=mydata; 
  tables agemos*agerand/list missing; 
run; 
 
proc means data=_cdcdata; 
  where treatment = 'GLARGINE + METFORMIN'; 
  var bmipct BMIZ; 
  title3 "Cohort: Glargine followed by metformin";  
run; 
 
proc means data=_cdcdata; 
  where treatment = 'METFORMIN'; 
  var bmipct BMIZ; 
  title3 "Cohort: metformin alone";  
run; 
 
 
data analy; 
  merge analy (in=in1) _cdcdata (in=in2 keep=RISE_REPOSITORY_ID bmipct bmiz); 
  by RISE_REPOSITORY_ID; 
  if in1 and in2; 
run; 
 
 
 
* Cohort: Glargine followed by metformin; 
        
* med, q1, q3; 
data prntunivg_m; 
 * length _VAR_ $100; 
  set _null_; 
run; 
 
%univ(2    , agerand        , treat_grp1 , G_M);    
%univ(8    , Weight         , treat_grp1 , G_M);   
%univ(9    , bmi1           , treat_grp1 , G_M);                  
%univ(9    , bmipct         , treat_grp1 , G_M);                   
%univ(9    , bmiz           , treat_grp1 , G_M);                   
%univ(10   , waist1         , treat_grp1 , G_M);  * Waist circumference (cm)                ; 
%univ(11   , hba1c          , treat_grp1 , G_M);                    
%univ(10   , HbA1c_mol_mmol , treat_grp1 , G_M);                   
%univ(16   , sbp            , treat_grp1 , G_M);                     
%univ(17   , dbp            , treat_grp1 , G_M);                     
*Triglycerides (mmol/L)                   ;   * (95% CI not calculating) ;  

%univ(20   , hdl_mmol_L     , treat_grp1 , G_M);                      
%univ(19   , ldl_mmol_L     , treat_grp1 , G_M);                      
%univ(12   , clamp_GLUC_05mod, treat_grp1 , G_M); * Hyperglycemic clamp: Fasting glucose (mmol/L)  ;                    
%univ(14   , CPEP_05_nmol_l , treat_grp1 , G_M);  * Hyperglycemic clamp: Fasting C-peptide (nmol/L);                    
* Fasting insulin (pmol/L)           ;   * (95% CI not calculating) ; 
* Steady-state C-peptide (nmol/L)    ;   * (95% CI not calculating) ; 
* ACPRmax (nmol/L)     ;   * (95% CI not calculating) ; 
* ACPRg (nmol/L)      ;   * (95% CI not calculating) ; 
%univ(14   , GIR            , treat_grp1 , G_M);  * Glucose infusion rate (M) (mmol/kg/min) ;   ** what is conversion??; 
* Steady-state insulin (I) (pmol/L)    ;   * (95% CI not calculating) ; 
* M/I (3 10e25 mmol/kg/min per pmol/L)   ;   * (95% CI not calculating) ; 
* OGTT: Fasting glucose (mmol/L)          ;    * (95% CI not calculating) ; 
%univ(12   , GLUC_05mod     , treat_grp1 , G_M);  * OGTT: Fasting glucose (mmol/L)  ;                   
%univ(15   , gluc120mod     , treat_grp1 , G_M);  * OGTT: 2-h glucose (mmol/L)     ; 
 
 
data prntunivg_m;     
  set prntunivg_m; 
   _median_  = round(_median_ , 0.1); 
   _q1_      = round(_q1_     , 0.1); 
   _q3_      = round(_q3_     , 0.1); 
   _mean_    = round(_mean_   , 0.001); 
   _std_     = round(_std_    , 0.001); 
run; 
 
 
proc print data=prntunivg_m; 
 var rownum _var_ covarf _nobs_/* _median_ _q1_ _q3_ _min_ _max_*/ _mean_ _std_; 
 title3 "Cohort: Glargine followed by metformin"; 
run; 
 
 
data prntg_m; 
 * length _VAR_ $100; 
  set _null_; 
run; 
 
%npercent(1, Sex          , SEXF   , treat_grp1, g_m);  
%npercent(2, scbrst_5     , ynf    , treat_grp1, g_m); *Tanner stage V, n (%) ;    
%npercent(3, Race         , RACEgF , treat_grp1, g_m);       * ??; 
%npercent(4, diabetes     , ynf    , treat_grp1, g_m); * IGT/Diabetes ; 
%npercent(5, SCMET        , ynf    , treat_grp1, g_m); * Metformin use at baseline, n (%); 
%npercent(6, hypertensive , ynf    , treat_grp1, g_m); * Hypertensive;    * ??; 
 
data prntg_m; 
  set prntg_m; 
  percent = round(percent,0.1); 
run; 

proc sort data=prntg_m; 
  by rownum covarf; 
  run; 
 
proc print data=prntg_m; 
  var rownum covar covarf count percent; 
run; 
 
 
* Cohort: metformin alone; 
        
* med, q1, q3; 
data prntunivmet; 
 * length _VAR_ $100; 
  set _null_; 
run; 
 
%univ(2    , agerand        , treat_grp3 , met);     
%univ(8    , Weight         , treat_grp3 , met);   
%univ(9    , bmi1           , treat_grp3 , met);                  
%univ(9    , bmipct         , treat_grp3 , met);                    
%univ(9    , bmiz           , treat_grp3 , met);                    
%univ(10   , waist1         , treat_grp3 , met);  * Waist circumference (cm)                ; 
%univ(11   , hba1c          , treat_grp3 , met);                    
%univ(10   , HbA1c_mol_mmol , treat_grp3 , met);                   
%univ(16   , sbp            , treat_grp3 , met);                     
%univ(17   , dbp            , treat_grp3 , met);                     
*Triglycerides (mmol/L)                   ;   * (95% CI not calculating) ;  
%univ(20   , hdl_mmol_L     , treat_grp3 , met);                      
%univ(19   , ldl_mmol_L     , treat_grp3 , met);                      
%univ(12   , clamp_GLUC_05mod, treat_grp3 , met); * Hyperglycemic clamp: Fasting glucose (mmol/L)  ;                    
%univ(14   , CPEP_05_nmol_l , treat_grp3 , met);  * Hyperglycemic clamp: Fasting C-peptide (nmol/L);                    
* Fasting insulin (pmol/L)           ;   * (95% CI not calculating) ; 
* Steady-state C-peptide (nmol/L)    ;   * (95% CI not calculating) ; 
* ACPRmax (nmol/L)     ;   * (95% CI not calculating) ; 
* ACPRg (nmol/L)      ;   * (95% CI not calculating) ; 
%univ(14   , GIR            , treat_grp3 , met);  * Glucose infusion rate (M) (mmol/kg/min) ;   ** what is conversion??; 
* Steady-state insulin (I) (pmol/L)    ;   * (95% CI not calculating) ; 
* M/I (3 10e25 mmol/kg/min per pmol/L)   ;   * (95% CI not calculating) ; 
* OGTT: Fasting glucose (mmol/L)          ;    * (95% CI not calculating) ; 
%univ(12   , GLUC_05mod     , treat_grp3 , met);  * OGTT: Fasting glucose (mmol/L)  ;                   
%univ(15   , gluc120mod     , treat_grp3 , met);  * OGTT: 2-h glucose (mmol/L)     ; 
 
 
data prntunivmet;     
  set prntunivmet; 
   _median_  = round(_median_ , 0.1); 
   _q1_      = round(_q1_     , 0.1); 
   _q3_      = round(_q3_     , 0.1); 

   _mean_    = round(_mean_   , 0.001); 
   _std_     = round(_std_    , 0.001); 
run; 
 
 
proc print data=prntunivmet; 
 var rownum _var_ covarf _nobs_/* _median_ _q1_ _q3_ _min_ _max_*/ _mean_ _std_; 
 title3 "Cohort: metformin alone"; 
run; 
 
 
data prntmet; 
 * length _VAR_ $100; 
  set _null_; 
run; 
 
%npercent(1, Sex          , SEXF   , treat_grp3, met);  
%npercent(2, scbrst_5     , ynf    , treat_grp3, met); *Tanner stage V, n (%) ;    
%npercent(3, Race         , RACEgF , treat_grp3, met);   
%npercent(4, diabetes     , ynf    , treat_grp3, met); * IGT/Diabetes ; 
%npercent(5, RSMET        , ynf    , treat_grp3, met); * Metformin use at baseline, n (%); 
%npercent(6, hypertensive , ynf    , treat_grp3, met); * Hypertensive; 
 
data prntmet; 
  set prntmet; 
  percent = round(percent,0.1); 
run; 
 
proc sort data=prntmet; 
  by rownum covarf; 
  run; 
 
proc print data=prntmet; 
  var rownum covar covarf count percent; 
run; 
 
 
options nocenter validvarname=upcase fmtsearch=(formats) nofmterr; 
 
title '/prj/niddk/ims_analysis/RISE/prog_initial_analysis/RISE_Pediatric_Medication/calculate_bmi_z_and_hypertension.sas'; 
run; 
 
* Peds primary outcome.pdf- calculate BMI percentile and z score using CDC site:  
https://www.cdc.gov/nccdphp/dnpao/growthcharts/resources/sas.htm;   
 
**********; 
* INPUT   ; 
**********; 
libname refdir '/prj/niddk/ims_analysis/RISE/private_created_data/'; 

libname orig   '/prj/niddk/ims_analysis/RISE/private_orig_data/RISE Pediatric Medication/';  
libname orig2  '/prj/niddk/ims_analysis/RISE/private_orig_data/RISE combined NIDDK only/';   
 
 
************; 
* MACROS    ; 
************; 
%macro readin(lib, ds); 
  data &ds; 
    set &lib..&ds; 
  run; 
 
  proc contents data=&ds; 
  title3 "&ds"; 
  run; 
%mend; 
 
 
%readin(refdir, cdcref_d);   
 
%readin(orig, risepeds_baseline);  
%readin(orig, risepeds_basedata);  
 
proc freq data=risepeds_basedata; 
  tables sex/missing; 
  run; 
 
%readin(orig2, risepeds_basedata_niddk_only); 
 
proc freq data=risepeds_basedata_niddk_only; 
  tables agerand sex*treatment/missing list; 
run; 
 
 
* create mydata with varnames specified in CDC code; 
proc sort data=risepeds_baseline; 
  by RISE_REPOSITORY_ID; 
run; 
 
proc sort data=risepeds_basedata_niddk_only; 
  by RISE_REPOSITORY_ID; 
run; 
 
data mydata (keep=RISE_REPOSITORY_ID height weight agerand agemos sex treatment bmi dbp sbp BAANTIHYP); 
  merge risepeds_baseline            (in=in1 keep=RISE_REPOSITORY_ID height1 weight1 dbp sbp BAANTIHYP 
                                             rename=(height1=height weight1=weight)) 
       risepeds_basedata_niddk_only (in=in2 keep=RISE_REPOSITORY_ID AGERAND sex treatment); 
  by RISE_REPOSITORY_ID; 
  if not (in1 and in2) then abort; 

  if in1 and in2; 
 
   bmi = weight/((height/100)**2); 
 
   agemos = agerand*12; 
 
   label agemos = "Age (months" 
         bmi    = "BMI (kg/m^2)" 
  ; 
 
   %include '/prj/niddk/ims_analysis/RISE/private_created_data/cdc-source-code.sas'; 
 
run; 
 
proc freq data=mydata; 
  tables agemos*agerand/list missing; 
run; 
 
proc means data=_cdcdata; 
  where treatment = 'GLARGINE + METFORMIN'; 
  var bmipct BMIZ; 
  title3 "Cohort: Glargine followed by metformin";  
run; 
 
proc means data=_cdcdata; 
  where treatment = 'METFORMIN'; 
  var bmipct BMIZ; 
  title3 "Cohort: metformin alone";  
run; 
 
*** Calculating Hypertension; 
 
 
data _indata;  
 set _cdcdata; 
 
 if 0<HAPCT<7.5 then ht_gp=1;  *** 5th %ile; 
 else if 7.5<=HAPCT<20 then ht_gp=2;  *** 10th %ile; 
 else if 20<=HAPCT<40 then ht_gp=3;  *** 25th %ile; 
 else if 40<=HAPCT<65 then ht_gp=4;  *** 50th %ile; 
 else if 65<=HAPCT<85 then ht_gp=5;  *** 75th %ile; 
 else if 85<=HAPCT<92.5 then ht_gp=6;  *** 90th %ile; 
 else if 92.5<=HAPCT<100 then ht_gp=7;  *** 95th %ile; 
  
 if SEX=1 then SEX_NUM=1;  
 else SEX_NUM=2; 
 
 
proc sort data = _indata;  

 by sex_num AGERAND; 
 
*** Boys; 
DATA BPB90 BPB95 BPB99;  **DATA FILE FOR BP-FOR-AGE -- boys; 
INFILE CARDS PAD missover; 
/* Age bp%ile sht5 sht10 sht25 sht50 sht75 sht90 sht95 
    dht5 dht10 dht25 dht50 dht75 dht90 dht95; 
*/     
     
input AGERAND bpptile sht1 sht2 sht3 sht4 sht5 sht6 sht7 
                       dht1 dht2 dht3 dht4 dht5 dht6 dht7; 
 
 
sex_num=1; 
if bpptile=90 then output bpb90; 
else if bpptile=95 then output bpb95; 
else if bpptile=99 then output bpb99; 
 
CARDS; 
 
10 90 111 112 114 115 117 119 119 73 73 74 75 76 77 78 
10 95 115 116 117 119 121 122 123 77 78 79 80 81 81 82 
10 99 122 123 125 127 128 130 130 85 86 86 88 88 89 90 
11 90 113 114 115 117 119 120 121 74 74 75 76 77 78 78 
11 95 117 118 119 121 123 124 125 78 78 79 80 81 82 82 
11 99 124 125 127 129 130 132 132 86 86 87 88 89 90 90 
12 90 115 116 118 120 121 123 123 74 75 75 76 77 78 79 
12 95 119 120 122 123 125 127 127 78 79 80 81 82 82 83 
12 99 126 127 129 131 133 134 135 86 87 88 89 90 90 91 
13 90 117 118 120 122 124 125 126 75 75 76 77 78 79 79 
13 95 121 122 124 126 128 129 130 79 79 80 81 82 83 83 
13 99 128 130 131 133 135 136 137 87 87 88 89 90 91 91 
14 90 120 121 123 125 126 128 128 75 76 77 78 79 79 80 
14 95 124 125 127 128 130 132 132 80 80 81 82 83 84 84 
14 99 131 132 134 136 138 139 140 87 88 89 90 91 92 92 
15 90 122 124 125 127 129 130 131 76 77 78 79 80 80 81 
15 95 126 127 129 131 133 134 135 81 81 82 83 84 85 85 
15 99 134 135 136 138 140 142 142 88 89 90 91 92 93 93 
16 90 125 126 128 130 131 133 134 78 78 79 80 81 82 82 
16 95 129 130 132 134 135 137 137 82 83 83 84 85 86 87 
16 99 136 137 139 141 143 144 145 90 90 91 92 93 94 94 
17 90 127 128 130 132 134 135 136 80 80 81 82 83 84 84 
17  95 131 132 134 136 138 139 140 84 85 86 87 87 88 89 
17 99 139 140 141 143 145 146 147 92 93 93 94 95 96 97 
; 
 
run; 

*** Girls; 
DATA BPG90 BPG95 BPG99 ;  **DATA FILE FOR BP-FOR-AGE -- GIRLS; 
INFILE CARDS PAD; 
/* Age bp%ile sht5 sht10 sht25 sht50 sht75 sht90 sht95 
    dht5 dht10 dht25 dht50 dht75 dht90 dht95;  */ 
     
input AGERAND bpptile sht1 sht2 sht3 sht4 sht5 sht6 sht7 
                       dht1 dht2 dht3 dht4 dht5 dht6 dht7; 
 
sex_num=2; 
if bpptile=90 then output bpg90; 
else if bpptile=95 then output bpg95; 
else if bpptile=99 then output bpg99; 
 
CARDS; 
 
10 90 112 112 114 115 116 118 118 73 73 73 74 75 76 76 
10 95 116 116 117 119 120 121 122 77 77 77 78 79 80 80 
10 99 123 123 125 126 127 129 129 84 84 85 86 86 87 88 
11 90 114 114 116 117 118 119 120 74 74 74 75 76 77 77 
11 95 118 118 119 121 122 123 124 78 78 78 79 80 81 81 
11 99 125 125 126 128 129 130 131 85 85 86 87 87 88 89 
12 90 116 116 117 119 120 121 122 75 75 75 76 77 78 78 
12 95 119 120 121 123 124 125 126 79 79 79 80 81 82 82 
12 99 127 127 128 130 131 132 133 86 86 87 88 88 89 90 
13 90 117 118 119 121 122 123 124 76 76 76 77 78 79 79 
13 95 121 122 123 124 126 127 128 80 80 80 81 82 83 83 
13 99 128 129 130 132 133 134 135 87 87 88 89 89 90 91 
14 90 119 120 121 122 124 125 125 77 77 77 78 79 80 80 
14 95 123 123 125 126 127 129 129 81 81 81 82 83 84 84 
14 99 130 131 132 133 135 136 136 88 88 89 90 90 91 92 
15 90 120 121 122 123 125 126 127 78 78 78 79 80 81 81 
15 95 124 125 126 127 129 130 131 82 82 82 83 84 85 85 
15 99 131 132 133 134 136 137 138 89 89 90 91 91 92 93 
16 90 121 122 123 124 126 127 128 78 78 79 80 81 81 82 
16 95 125 126 127 128 130 131 132 82 82 83 84 85 85 86 
16 99 132 133 134 135 137 138 139 90 90 90 91 92 93 93 
17 90 122 122 123 125 126 127 128 78 79 79 80 81 81 82 
17 95 125 126 127 129 130 131 132 82 83 83 84 85 85 86 
17 99 133 133 134 136 137 138 139 90 90 91 91 92 93 93 
; 
  
 
RUN; 
proc sort data=bpb90; by AGERAND; run; 
proc sort data=bpb95; by AGERAND; run; 
proc sort data=bpb99; by AGERAND; run; 

data bpb; merge bpb90(rename=(sht1=sht901 sht2=sht902 sht3=sht903 
                        sht4=sht904 sht5=sht905 sht6=sht906 sht7=sht907 
                              dht1=dht901 dht2=dht902 dht3=dht903 
                        dht4=dht904 dht5=dht905 dht6=dht906 dht7=dht907)) 
                bpb95(rename=(sht1=sht951 sht2=sht952 sht3=sht953 
                        sht4=sht954 sht5=sht955 sht6=sht956 sht7=sht957 
                              dht1=dht951 dht2=dht952 dht3=dht953 
                        dht4=dht954 dht5=dht955 dht6=dht956 dht7=dht957)) 
                bpb99(rename=(sht1=sht991 sht2=sht992 sht3=sht993 
                        sht4=sht994 sht5=sht995 sht6=sht996 sht7=sht997 
                              dht1=dht991 dht2=dht992 dht3=dht993 
                        dht4=dht994 dht5=dht995 dht6=dht996 dht7=dht997)); 
  by AGERAND; 
  run; 
 
 
proc sort data=bpg90; by AGERAND; run; 
proc sort data=bpg95; by AGERAND; run; 
proc sort data=bpg99; by AGERAND; run; 
  
  
data bpg; merge bpg90(rename=(sht1=sht901 sht2=sht902 sht3=sht903 
                        sht4=sht904 sht5=sht905 sht6=sht906 sht7=sht907 
                              dht1=dht901 dht2=dht902 dht3=dht903 
                        dht4=dht904 dht5=dht905 dht6=dht906 dht7=dht907)) 
                  
                bpg95(rename=(sht1=sht951 sht2=sht952 sht3=sht953 
                        sht4=sht954 sht5=sht955 sht6=sht956 sht7=sht957 
                              dht1=dht951 dht2=dht952 dht3=dht953 
                        dht4=dht954 dht5=dht955 dht6=dht956 dht7=dht957)) 
                 
                bpg99(rename=(sht1=sht991 sht2=sht992 sht3=sht993 
                        sht4=sht994 sht5=sht995 sht6=sht996 sht7=sht997 
                              dht1=dht991 dht2=dht992 dht3=dht993 
                       dht4=dht994 dht5=dht995 dht6=dht996 dht7=dht997)); 
  by AGERAND; 
  run; 
 
 
*** Allow max 95%ile of 130/80 and max 90%ile of 125/75; 
  ** do not have 99th percentile cutpoint - use 130/85; 
data bpall;  
 set bpb  
   bpg;  
  array sbp90(7) sht901--sht907; 
  array sbp95(7) sht951--sht957; 
  array sbp99(7) sht991--sht997; 

  array dbp90(7) dht901--dht907; 
  array dbp95(7) dht951--dht957; 
  array dbp99(7) dht991--dht997; 
   
do i=1 to 7; ** do not have 99th percentile cutpoint - use 130/85; 
  if sbp90(i)>125 then sbp90(i)=125; 
  if sbp95(i)>130 then sbp95(i)=130; 
  if sbp99(i)>130 then sbp99(i)=130; 
   
   
  if dbp90(i)>75 then dbp90(i)=75; 
  if dbp95(i)>80 then dbp95(i)=80; 
  if dbp99(i)>85 then dbp99(i)=85; 
end; 
run; 
 
proc sort data=bpall;  
 by sex_num AGERAND; 
 
 
proc sort data = bpall NODUPKEY; 
 by sex_num AGERAND;  
 
proc print data = bpall; 
 title3 "bpall"; 
 
 
data newdata;  
 merge _indata (in=in1)  
     bpall;  
 by sex_num AGERAND; 
   
  array sbp90(7) sht901--sht907; 
  array sbp95(7) sht951--sht957; 
  array sbp99(7) sht991--sht997; 
   
  array dbp90(7) dht901--dht907; 
  array dbp95(7) dht951--dht957; 
  array dbp99(7) dht991--dht997; 
 
  if ht_gp ne . and AGERAND<=17 then do; 
    
    if sbp>=sbp90(ht_gp) then hisbp90=1; else hisbp90=0; 
    if sbp>=sbp95(ht_gp) then hisbp95=1; else hisbp95=0; 
    if sbp>=sbp99(ht_gp) then hisbp99=1; else hisbp99=0; 
     
    if dbp>=dbp90(ht_gp) then hidbp90=1; else hidbp90=0; 
    if dbp>=dbp95(ht_gp) then hidbp95=1; else hidbp95=0; 
    if dbp>=dbp99(ht_gp) then hidbp99=1; else hidbp99=0; 

    sbp90v=sbp90(ht_gp); 
    sbp95v=sbp95(ht_gp);  
    sbp99v=sbp99(ht_gp);  
     
    dbp90v=dbp90(ht_gp); 
    dbp95v=dbp95(ht_gp); 
    dbp99v=dbp99(ht_gp); 
     
 if hisbp90=1 or hidbp90=1 or BAANTIHYP=1 then hibp90=1;  
  else hibp90=0; 
   
 if hisbp95=1 or hidbp95=1 or BAANTIHYP=1 then hibp95=1;  
   else hibp95=0; 
    
 if hisbp99=1 or hidbp99=1 or BAANTIHYP=1 then hibp99=1;  
   else hibp99=0; 
 
  end; 
  
  if AGERAND>=18 then do;  *** Use adult values; 
    if sbp>125 then hisbp90=1;  
      else hisbp90=0; 
    if sbp>130 then hisbp95=1;  
      else hisbp95=0; 
    if sbp>130 then hisbp99=1;  
      else hisbp99=0; 
     
    if dbp>75 then hidbp90=1;  
      else hidbp90=0; 
    if dbp>80 then hidbp95=1;  
      else hidbp95=0; 
    if dbp>85 then hidbp99=1;  
      else hidbp99=0; 
     
   if hisbp90=1 or hidbp90=1 or BAANTIHYP=1 then hibp90=1;  
     else hibp90=0; 
    if hisbp95=1 or hidbp95=1 or BAANTIHYP=1 then hibp95=1;  
      else hibp95=0; 
    if hisbp99=1 or hidbp99=1 or BAANTIHYP=1 then hibp99=1;  
      else hibp99=0; 
     
  end; 
   
  if in1 then output newdata; 
 
label   ht_gp='Height % grouping' 
        sbp90v='90th %ile cutpoint, SBP' 
        sbp95v='95th %ile cutpoint, SBP'  

        dbp90v='90th %ile cutpoint, DBP' 
        dbp95v='95th %ile cutpoint, DBP' 
        hisbp90='High SBP at 90th %ile or 125' 
        hisbp95='High SBP at 95th %ile or 130' 
        hisbp99='High SBP at 99th %ile or 130' 
        hidbp90='High DBP at 90th %ile or 75' 
        hidbp95='High DBP at 95th %ile or 80' 
        hidbp99='High DBP at 99th %ile or 85' 
        hibp90='High BP at 90th %ile or 125/75' 
        hibp95='High BP at 95th %ile or 130/80' 
        hibp99='High BP at 99th %ile or 130/85' 
        ;      
         
run; 
 
proc freq data = newdata; 
 table hibp90*hibp95*hibp99*BAANTIHYP / list missing; 
 
proc means data=_cdcdata; 
  where treatment = 'GLARGINE + METFORMIN'; 
  var bmipct BMIZ; 
  title3 "Checking BMI Percentage and BMIZ in Table 1 of the Publication (Cohort: Glargine followed by metformin)";  
run; 
 
proc means data=_cdcdata; 
  where treatment = 'METFORMIN'; 
  var bmipct BMIZ; 
  title3 "Checking BMI Percentage and BMIZ in Table 1 of the Publication (Cohort: metformin alone)";  
run; 
 
proc freq data = newdata; 
 table treatment*hibp95 / list missing; 
 where treatment = 'GLARGINE + METFORMIN'; 
 title3 "Checking the Hypertension Variable in Table 1 of the Publication (Cohort: Glargine followed by metformin)"; 
 
proc freq data = newdata; 
 table treatment*hibp95 / list missing; 
 where treatment = 'METFORMIN'; 
 title3 "Checking the Hypertension Variable in Table 1 of the Publication (Cohort: metformin alone)";  

